Subject: Re: fast array comparison
Posted by David Fanning on Mon, 09 Dec 2002 16:00:38 GMT
View Forum Message <> Reply to Message

JD Smith (jdsmith@as.arizona.edu) writes:

> This topic was discussed ad naseum over the past couple of years, with
> the critical differences between looking for the values and looking for
> the indices of intersection pointed out. Several different methods were
> compared using HISTOGRAM, SORT, and direct array inflation. Depending on
> your problem size, one of these will be fastest. Usually. ;).
>
> Give it a search on Gusenet.

JD hasn't had his coffee yet this morning. :-)

Search the IDL newsgroup archives on Goggle for "Matching Lists"
by Mark Fardal for the ad nauseam discussion JD mentions. You
may learn more about list searching than you ever wanted to know. :-)

Or, if you just want an answer (no sense reading this
newsgroup if you fall into this category), here you go:

```
;************************************************************ ******
FUNCTION SetIntersection, a, b, Indices=indices, Count=count
minab = Min(a, Max=maxa) > Min(b, Max=maxb) ; Intersection of ranges
maxab = maxa < maxb

  ; If either set is empty, or ranges don't intersect: result = NULL.

IF maxab LT minab OR maxab LT 0 THEN RETURN, -1
hist_a = Histogram(a, Min=minab, Max=maxab, Reverse_Indices=rev_a)
hist_b = Histogram(b, Min=minab, Max=maxab)
r = Where((hist_a NE 0) AND (hist_b NE 0), count)
IF count EQ 0 THEN BEGIN
   RETURN, -1
ENDIF ELSE BEGIN
   IF Arg_Present(indices) THEN indices = rev_a[rev_a[r]]
   RETURN, r + minab
ENDELSE
END
;************************************************************ ******

  IDL> request_array = [5,6,7,8,9,10]
  IDL> avail_array = [3,7,8,9,12,13,16]
  IDL> int = setintersection(avail_array, request_array, Indices=I)
  IDL> print, int, I
         7         8         9
```

1 2 3

Remember, this is one of three possible algorithms discussed in that series of articles. Choose wisely.

Cheers,

David

--
David W. Fanning, Ph.D.
Fanning Software Consulting, Inc.
Phone: 970-221-0438, E-mail: david@dfanning.com
Coyote's Guide to IDL Programming: http://www.dfanning.com/
Toll-Free IDL Book Orders: 1-888-461-0155
